



# Index Embosser Interface Protocol

For V4 Embossers

Creating Temporary Paper and Label Definitions

*Version 0.2*

Index Braille © 2014

May 19, 2014

[developer\\_support@indexbraille.com](mailto:developer_support@indexbraille.com)

## Contents

<b>1</b>	<b>Introduction</b>	<b>3</b>
<b>2</b>	<b>Syntax for Creating Temporary Paper and Label Definitions</b>	<b>3</b>
2.1	Syntax for Creating a Temporary Paper Definition . . . . .	4
<b>3</b>	<b>Syntax for Creating Temporary Label Definitions</b>	<b>5</b>

# 1 Introduction

This document describes how to create temporary paper and label definitions using [ESC]D sequences for Index Braille V4 embossers. The functionality is planned to be part of firmware release 1.5.3. *Please note that this document is not yet finalized and may be updated in the future.*

Temporary paper definitions can be created by on their own. However, it is necessary to create both a paper definition and a label definition when creating temporary label definitions.

## 2 Syntax for Creating Temporary Paper and Label Definitions

The syntax for defining a temporary paper definition differs somewhat from previous [ESC]D sequences. The new syntax can be recognized by the use of quotation marks (" "), directly following [ESC]D. The main reason is to provide a more flexible and powerful definition notation. It allows for variable length parameter names, instead of each parameter limited to having “two character”-names. All characters in the temporary paper and label definitions should be encoded using the *ASCII character encoding scheme* [2].

Both the creation of temporary paper definitions and temporary label definitions follow the same basic format. The sequence is started by the ASCII characters [ESC]D (hexadecimal: 1B 44). This is directly followed by a command, enclosed in quotation marks<sup>1</sup>, e.g. “**define paper**”, which is the command for temporarily defining a paper. This command is then directly followed by a list of parameters “⟨parameter-list⟩”, also enclosed in quotes, which contain the necessary parameters and their assigned values to fully define either a paper or a label. Please note, that the order of the parameters must follow the specification as there can exist dependencies between different parameters and the order of each specification ensures that such dependencies are handled properly.

The syntax is specified using a EBNF notation with the following typographical rules:

- Text that is in **bold face** are literals (terminals in EBNF) and should be sent to the embosser “as is”. Literals do not contain spaces, but can contain hyphens (hexadecimal: 2D).
- Text between apostrophes ’ (hexadecimal: 27) denote character strings and may contain all printable ASCII characters, see [1]. The only exceptions are the quotation mark “ (hexadecimal: 22) and the backslash \ (hexadecimal: 5C), which may not be included in character strings.
- Text between angled brackets ⟨⟩ are “symbols” (non-terminals in EBNF) that need further defining, .e.g., lists of parameters, values assigned to parameters.
- Text in boxes are non-formal descriptions of commonly known concepts, e.g. character string, decimal value. This is used to make the notation less cumbersome while (hopefully) remaining unambiguous. In several cases there are

<sup>1</sup>Quotations marks are hexadecimal: 22.

limits to the values that can be assigned to the parameters, e.g., max/min values for decimal values and max length of string descriptions. Existing limits are set in subscript, directly following the corresponding non-formal description. E.g., `character string`<sub>max-length=29</sub>, species that the max length for the string is 29 characters.

- Text enclosed in brackets [] are optional symbols or literals, that are only present in a definition if certain conditions are met.
- Text enclosed in curly brackets {} are used group symbols and/or literals. An asterisk behind a group, {}, indicates that the values can be repeated zero or more times. A question mark behind a group {}, indicates that the values can be repeated zero or one time.
- Text enclosed in (\*\*) are comments and used to clarify concepts.

## 2.1 Syntax for Creating a Temporary Paper Definition

The syntax to define a temporary paper definition, expressed in EBNF, is as follows:

`<paper-definition> ::= [ESC]D “define-paper ” “<paper-parameters>”`

`<paper-parameters> ::= <mandatory-list> [, <tractor-list>] [, <orientation-value>]`

`<mandatory-list> ::= description: ‘<paper-description>’, paper-length: <size>, paper-width: <size>, size-unit: <size-unit>, feed-type: <feed-type>`

`<tractor-list> ::= ribbon-width: <ribbon-size>, hole-count: <hole-count> [, repeat-hole-count: <repeat-count>]`  
 (\* only present if feed-type is tractor \*)

`<orientation-value> ::= load-orientation: <orientation>`  
 (\* only needs to be defined for landscape. If undefined, default is portrait. \*)

`<paper-description> ::= character stringmax-length=29`  
 (\* Describes the paper for presentation purposes. \*)

`<size> ::= decimal valuemm-max-value=2600.0 or inch-max-value=102.0`  
 (\* The maximum value depends on the unit size. \*)

`<ribbon-size> ::= decimal valuemax-value=paper width`

$\langle size-unit \rangle ::= \mathbf{mm}$   
 $\quad \quad \quad | \mathbf{inch}$   
 (\*e.g., allows for usage of mm or inch in paper size .\*)

$\langle feed-type \rangle ::= \mathbf{sheet}$   
 $\quad \quad \quad | \mathbf{tractor}$   
 (\*e.g., allows for usage of mm or inch in paper size .\*)

$\langle hole-count \rangle ::= \boxed{\text{integer value}}_{\text{max-value}=65535, \text{min-value}=0}$   
 (\* The number of holes on tractor feed page \*)

$\langle repeat-count \rangle ::= \boxed{\text{integer value}}_{\text{max-value}=65535, \text{min-value}=0}$   
 (\* Only needs to be defined if the tractor feed hole patterns doesn't repeat every page \*)

$\langle orientation \rangle ::= \mathbf{portrait}$   
 $\quad \quad \quad | \mathbf{landscape}$   
 (\* Default is portrait \*)

### 3 Syntax for Creating Temporary Label Definitions

The syntax to define a temporary label definition, expressed in EBNF, is as follows:

$\langle label-definition \rangle ::= [\mathbf{ESC}] \mathbf{D} \mathbf{"define-label"} \mathbf{"} \langle label-parameters \rangle \mathbf{"}$

$\langle label-parameters \rangle ::= \langle mandatory-list \rangle [ , \langle rotation-values \rangle ] [ , \langle margin-value \rangle ]$

$\langle mandatory-list \rangle ::= \mathbf{paper-select:} \langle paper-id \rangle, \mathbf{label-size-x:} \langle size \rangle, \mathbf{label-size-y:} \langle size \rangle,$   
 $\mathbf{size-unit:} \langle size-unit \rangle, \mathbf{number-of-labels:} \langle number-of-labels \rangle, \mathbf{label-origos:} \langle label-origo-list \rangle$

$\langle paper-id \rangle ::= \mathbf{custom-paper}$   
 $\quad \quad \quad | \boxed{\text{integer value}}$   
 (\* An integer value can be used to select all predefined paper types that are available in the embosser, e.g. A4, A3, letter. If custom paper is selected it must have been defined before the label definition\*)

$\langle rotation-values \rangle ::= \mathbf{label-rotations:} \langle rotation-list \rangle$

$\langle \text{margin-value} \rangle ::= \mathbf{x-margin:} \langle \text{size} \rangle, \mathbf{y-margin:} \langle \text{size} \rangle$   
 (\* Determines the margins for all labels \*)

$\langle \text{size} \rangle ::= \boxed{\text{decimal value}}$   
 (\*e.g., 210.0 or 15.755 or 23 .\*)

$\langle \text{size-unit} \rangle ::= \mathbf{mm}$   
 |  $\mathbf{inch}$   
 (\*e.g., allows for usage of mm or inch in paper size .\*)

$\langle \text{number-of-labels} \rangle ::= \boxed{\text{integer value}}$   
 (\* The number of labels on a page.\*)

$\langle \text{label-origo-list} \rangle ::= \{ \langle \text{origo-value} \rangle \}^? \{ \# \langle \text{origo-value} \rangle \}^*$   
 (\* Each origo value is separated by a number sign # (hexadecimal:23). The list must contain as many origo-values as there are labels on a page\*)

$\langle \text{origo-value} \rangle ::= \boxed{\text{decimal value}} \& \boxed{\text{decimal value}}$   
 (\* One x-value followed by one y-value. The decimal vales are separated by an ampersand & (hexadecimal: 26), specifying the origo of each label, e.g., 1.5&2.5 \*)

$\langle \text{rotation-list} \rangle ::= \{ \langle \text{rotate-value} \rangle \}^? \{ \# \langle \text{rotate-value} \rangle \}^*$   
 (\* Each rotation value is separated by a number sign # (hexadecimal: 23). if the list is defined, it must contain as many rotation-values as there are labels on a page. If the list is not defined, all labels are set to rotate-00 \*)

$\langle \text{rotate-value} \rangle ::= \mathbf{rotate-00}$   
 |  $\mathbf{rotate-90}$   
 |  $\mathbf{rotate-180}$   
 |  $\mathbf{rotate-270}$   
 (\* Specifies the rotation of each label\*)

## References

- [1] ASCII printable characters. [http://en.wikipedia.org/wiki/ASCII#ASCII\\_printable\\_characters](http://en.wikipedia.org/wiki/ASCII#ASCII_printable_characters).
- [2] ASCII table from Wikipedia. <http://en.wikipedia.org/wiki/ASCII>.